# $Id$
# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: Andrea Scarpino <andrea@archlinux.org>

pkgname=libkdegames4
pkgver=14.12.3
pkgrel=3
pkgdesc="Common code and data for many KDE games"
url="https://projects.kde.org/projects/kde/kdegames/libkdegames"
arch=('i686' 'x86_64')
license=('GPL' 'LGPL' 'FDL')
depends=('kdelibs' 'openal')
makedepends=('cmake' 'automoc4')
conflicts=('libkdegames<15')
source=("http://download.kde.org/Attic/applications/${pkgver}/src/libkdegames-${pkgver}.tar.xz")
sha1sums=('258d8130e7a38d1366466cef1d29ee2e13a456a5')

build() {
  mkdir build
  cd build
  cmake ../libkdegames-${pkgver} \
    -DCMAKE_BUILD_TYPE=Release \
    -DKDE4_BUILD_TESTS=OFF \
    -DCMAKE_INSTALL_PREFIX=/usr
  make
}

package() {
  cd build
  make DESTDIR="${pkgdir}" install
}
